

MATH 2415

CALCULUS III

INTRODUCING MAPLE

A GUIDED TUTORIAL FOR CALCULUS APPLICATIONS

TABLE OF CONTENTS

- Part I: Graphing in Two Dimensions, page 2
- Part II: Factoring, page 5
- Part III: Solving Equations, page 6
- Part IV: Limits, page 6
- Part V: Derivatives, page 7
- Part VI: Integrals, page 8
- Part VII: The Student Calculus Package, page 9
- Part VIII: Graphing Equations in Polar Form, page 15
- Part IX: Vectors and Vector Operations, page 16
- Part X: Graphing Vector Functions, page 18
- Part XI: Vector Calculus, page 20
- Part XII: Graphing in Three Dimensions, page 22

Note: Depending on how your copy of Maple is configured, the command lines may or may not begin with the prompt $>$. If you do not see that prompt, you don't have to type it in. Just type the remainder of the command.

MATH 2415 INTRODUCING MAPLE

Note: Most Maple commands end with a semi-colon. Subroutine calls end with a colon. The instructions below show how commands can be typed in with “DOS-like” syntax. The software will convert what you type to more mathematical looking format. You may need to use the right arrow key (\rightarrow) to get yourself “out” of an exponent or “out” of a fraction or radical. Some of the function commands (e.g. `sqrt`) can be replaced by picking a template out of the “Expression” menu on the left side of your screen. Play!

PART I: GRAPHING IN TWO DIMENSIONS


GRAPHING FUNCTIONS OF ONE VARIABLE

General Form: `plot (expression,variable=a..b) ;`

Example 1: Sketch the graph of $y = \sqrt{75 - 3x^2}$.

Note: You must define a domain for every function. Note the two dots in the domain definition.

`> plot (sqrt (75-3*x^2) , x=-5..5) ;`




Example 2: Sketch the graph of $y = \frac{2x}{x^2 - 4}$.

Note: It is optional to define a range for the function.

`> plot (2*x / (x^2-4) , x=-4..4 , y=-8..8) ;`

Click on the graph, go to the “plot” pull down menu and click on “scaling constrained” to see the graph with uniform spacing on both axes.



GRAPHING SEVERAL FUNCTIONS OF ONE VARIABLE


General Form: `plot({expr.1,expr.2,...,expr.n}, variable=a..b);`

Example 3: Sketch the graph of $x^2 + y^2 = 25$.

Note: The main part of the Maple program can only graph functions, but this relation can be replaced by two functions:

$$y = \pm \sqrt{25 - x^2}.$$

`> plot({sqrt(25-x^2), -sqrt(25-x^2)}, x=-5..5);`



(Scaling Constrained)

GRAPHING EQUATIONS THAT DO NOT DEFINE FUNCTIONS

Graphing equations that do not define functions is hard using a graphing calculator and not so easy using the regular “plot command in Maple.” Fortunately, there is a subroutine called “plots”. Not a part of the main Maple program, this subroutine is easy to call up nevertheless. This subroutine defines more sophisticated graphing commands including one called “implicitplot” that easily sketches graphs of equations whether they define functions or not.

If you want to use the subroutine, you must first invoke it. Fortunately, you only have to invoke it once in your Maple session and it is then available throughout your session. Enter the following, noting that, when invoking subroutines, you end the command with a colon rather than a semicolon.

`> with(plots);`


Graphing using “implicitplot”

General Form: `implicitplot(equation, x=a..b, y=c..d);`

Notice that you must specify intervals of values for both x and y . Be willing to play with those.


Example 4: Sketch the graph of $y^2 - 4y - 4x - 8 = 0$.

```
> implicitplot(y^2-4*y-4*x-8=0, x=-3..3, y=-4..7);
```



Example 5: Sketch the graph of $9x^2 + 18x + 16y = 4y^2 + 43$.

```
> implicitplot(9*x^2+18*x+16*y=4*y^2+43, x=-8..6, y=-8..10);
```



PART II: FACTORING

Maple can factor a polynomial over just the rational numbers or over all complex numbers.

FACTORING OVER THE RATIONAL NUMBERS

General Form: `factor (polynomial) ;`

Example 6: Factor the polynomial $15x^2 + 2x - 8$.

```
> factor (15*x^2+2*x-8) ;
(5x+4)(3x-2)
```

Example 7: Factor the polynomial $x^2 + 4$.

```
> factor (x^2+4) ;
x^2 + 4.
```

This polynomial doesn't factor over the rational numbers (i.e. with rational number coefficients).

FACTORING OVER THE COMPLEX NUMBERS

General Form: `factor (polynomial,complex) ;`

Note: This command will return a factorization in the form $a(x-b)(x-c)$.

Example 8: Factor the polynomial using this command. $x^2 + 4$

```
> factor (x^2+4, complex) ;
(x+2.000000000I)(x-2.000000000I)
```

Note that $I = i$. The factorization here is $(x + 2i)(x - 2i)$.

Example 9: Factor the polynomial using this command. $15x^2 + 2x - 8$.

```
> factor (15*x^2+2*x-8, complex) ;
15.(x+0.8000000000)(x-0.666666667)
```

PART III: SOLVING EQUATIONS

Solving Equations

General Form: `solve (equation) ;`

Example 10: Solve the equation $40x^4 - 18x^3 - 58x^2 + 27x - 3 = 0$.

> `solve (40*x^4-18*x^3-58*x^2+27*x-3=0) ;`
 $\frac{1}{4}, \frac{1}{5}, \frac{1}{2}\sqrt{6}, -\frac{1}{2}\sqrt{6}$

Example 11: Find the roots of the function $f(x) = 2x^4 - 11x^3 + 11x^2 + 15x - 9$.

In other words, solve the equation $2x^4 - 11x^3 + 11x^2 + 15x - 9 = 0$

> `solve (2*x^4-11*x^3+11*x^2+15*x-9=0) ;`
 $-1, \frac{1}{2}, 3, 3$

Note that repeated roots are listed according to their multiplicity.

PART IV: LIMITS

Maple can evaluate limits of functions, including one sided limits.

Two-Sided Limits

General Form: `limit (function, variable=a) ;`

Example 12: Find $\lim_{x \rightarrow 1} \frac{x^2 - 2x + 1}{x^4 + 3x^3 - 7x^2 + x + 2}$.

> `limit ((x^2-2*x+1) / (x^4+3*x^3-7*x^2+x+2), x=1) ;`
 $\frac{1}{8}$

One-Sided Limits

General Form: `limit (function, variable=a, left or right) ;`

Example 13: Find $\lim_{x \rightarrow 2^-} \frac{3x^2 + 5}{x^2 - 4}$.

```
> limit((3*x^2+5)/(x^2-4), x=2, left);
      -∞
```

PART V: DERIVATIVES

Maple will find derivatives of functions as long as you make explicit with respect to which variable you are differentiating.

General Form: `diff (function,variable) ;`

Example 14: Let $f(x) = \sin(2x) + 2x \cos(2x) + 2x$. Find $f'(x)$.

Note that you are differentiating with respect to x .

```
> diff(sin(2*x)+2*x*cos(2*x)+2*x, x);
      4cos(2x)-4xsin(2x)+2
```

Example 15: Given the function defined in Example 14, find $f'\left(\frac{\pi}{2}\right)$.

The command *eval* allows you to evaluate a given function at a given value. In Maple, % will call up the result of the most recent operation. So, if you just finished Example 14, rather than typing in the derivative you can type:

```
> eval(%, x=Pi/2);
      -2
```

Example 16: Let $f(x) = x^4 - 12x^3 + 48x^2 - 64x$. Find all values of x at which the graph of f has horizontal tangent lines.

```
> diff(x^4-12*x^3+48*x^2-64*x, x);
      4x^3-36x^2+96x-64
```

```
> solve(%=0);
      1, 4, 4
```

PART VI: INTEGRALS

Maple can find both indefinite and definite integrals of functions as long as the functions aren't too complicated.

Indefinite Integrals

You must specify respect to which variable you are integrating.

General Form: `int (function, variable) ;`

Example 17: Evaluate $\int x \sin x \, dx$.

```
> int(x*sin(x), x);
sin(x) - x*cos(x)
```

(Notice that Maple forgets to include the "+C". Shame! Shame!

Example 18: Evaluate $\int \frac{x}{\sqrt{x^2 - 6x + 5}} \, dx$.

```
> int(x/(sqrt(x^2-6*x+5)), x);
sqrt(x^2-6x+5) + 3*ln(x-3+sqrt(x^2-6x+5))
```

Definite Integrals

General Form: `int (function, variable=a..b) ;`

Example 19: Evaluate $\int_0^1 \ln(1+t^2) \, dt$.

```
> int(ln(1+t^2), t=0..1);
ln(2) - 2 + 1/2*pi
```

PART VII: THE STUDENT CALCULUS PACKAGE

Many more advanced and elaborate Calculus operations can be performed using the Student Calculus Package, a subroutine of Maple you can call. We will explore this package more throughout the next three semesters. As an example we will show how this package can illustrate tangent lines to curves.

Like the plots subroutine, you have to call the Student Calculus Package before you can use it, but you only have to call it once per Maple session. As before, subroutine calls end with a colon rather than a semicolon.

To call this package, type:


```
> with(student) :
```

SHOWING A TANGENT LINE TO A CURVE

General Form: `showtangent (function, x= x coordinate of point of tangency) ;`

Example 20: Sketch the graph of a tangent line to the graph of the function given by $f(x) = x^2 + 5$ at $x = 2$.

```
> showtangent (x^2+5, x=2) ;
```




ILLUSTRATING AN AREA USING RIGHT ENDPOINT RECTANGLES

General Form: `rightbox (function,variable=a..b,number of rectangles) ;`

Example 21: Illustrate the area bounded by the graph of $y = 5x - x^2$ and the x axis using 10 right endpoint rectangles.

First, we sketch the graph and calculate the x intercepts.


```
> plot(5*x-x^2, x=-1..6);
```



```
> solve(5*x-x^2=0);
0, 5
```

Now we use the `rightbox` command.

```
> rightbox(5*x-x^2, x=0..5, 10);
```



CALCULATING A RIEMANN SUM OF THE AREAS OF RIGHT ENDPPOINT RECTANGLES

General Form: `rightsum (function, variable=a..b,number of rectangles);`

Example 22: Calculate the total area of the rectangles in Example 21.

$$\begin{aligned} &> \text{rightsum}(5*x-x^2, x=0..5, 10); \\ &\quad \frac{1}{2} \left(\sum_{i=1}^{10} \left(\frac{5}{2}i - \frac{1}{4}i^2 \right) \right) \end{aligned}$$

Notice that this command expresses the total area as a sum. To evaluate the sum enter:

$$\begin{aligned} &> \text{value}(\%); \\ &\quad \frac{165}{8} \end{aligned}$$

Example 23: Find the area under this curve exactly using a definite integral.

$$\begin{aligned} &> \text{int}(5*x-x^2, x=0..5); \\ &\quad \frac{125}{6} \end{aligned}$$

SOLIDS OF REVOLUTION

Maple can take either a region bounded by the graph of a function and the x axis or a region bounded by the graphs of two functions, between two specified values of x , and rotate that region around either a horizontal or a vertical axis.

Limitations: The bounding curves have to be functions defined explicitly. There can only be two of them.

This capability is part of a subroutine you have to call. Type the following once and it is active throughout your Maple session.

```
> with(Student[Calculus1]):
```

General Format: `> VolumeOfRevolution(f(x), x=a..b, options);`
or: `> VolumeOfRevolution(f(x), g(x), x=a..b, options);`

In the options area you can type any of the following; you can have more than one.

Options:

`axis` = *horizontal or vertical* (whether the expression is rotated around a horizontal axis (like around the x axis) or a vertical axis (like around the y axis))

`distancefromaxis` = *numeric* (the distance from the coordinate axis to the line of rotation. For example,

`axis=horizontal, distancefromaxis=-2`
rotates around the line $y = -2$. If you don't write in a `distancefromaxis` option, rotation will be around a coordinate axis.

`output` = *value, integral or plot*

`plot` means the rotated solid will be graphed.

`integral` means that an integral will be written that would give the volume of the solid.

`value` means that the actual volume will be figured.


Example 24: Sketch the region in the plane bounded by the graphs of

$$y = x^2 + 1, \quad y = 0, \quad x = 0, \quad x = 3.$$

Note: Since the last two of these equations don't define functions, you will need to use "implicit plot." If you have not done so since the last time you opened Maple you first have to call a subroutine.


> with(plots):

> implicitplot({ $y=x^2+1, y=0, x=0, x=3$ }, $x=-1..4, y=-1..11$);



Example 25: Suppose the region bounded by the graphs of
 $y = x^2 + 1$, $y = 0$, $x = 0$, $x = 3$.
 is rotated around the x axis. Show the resulting solid.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=horizontal,output=plot);
```



Important Note: “Grab” the figure with your cursor. You can move the figure around to see it from any angle. Try this for a while!!!

Example 26: Set up an integral to calculate the volume of the figure in Example 25.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=horizontal,output=integral);
```

$$\int_0^3 \pi (x^2 + 1)^2 dx$$

Example 27: Find this volume.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=horizontal,output=value);
```

$$\frac{348}{5} \pi$$


Example 28: Suppose the region bounded by the graphs of

$$y = x^2 + 1, \quad y = 0, \quad x = 0, \quad x = 3.$$

is now rotated around the y axis. Show the resulting solid.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=vertical,output=plot);
```

The Volume of Revolution Around the Vertical Axis of
 $f(x) = x^2 + 1$
 on the Interval $[0, 3]$



Example 29: Set up an integral to calculate the volume of the figure in Example 28.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=vertical,output=integral);
```

$$\int_0^3 2 \pi x (x^2 + 1) dx$$

Example 30: Find this volume.

```
>VolumeOfRevolution(x^2+1,x=0..3,axis=vertical,output=value);
```

$$\frac{99}{2} \pi$$

PART VIII. GRAPHING EQUATIONS IN POLAR FORM

GRAPHING A SINGLE POLAR EQUATION

Note: Graphing polar equations is part of the plots subroutine. If you have already opened that subroutine in your current Maple session, you don't have to do so again. If not, first enter:


```
> with(plots):
```

General Form: `plots[polarplot](expression, variable=a..b);`

Example 31: Sketch the graph of $r = 1 - \sin \theta$.

Notes: You have to choose only Latin letters for variables; π is represented by Pi.

```
>plots[polarplot](1-sin(t), t=0..2*Pi);
```




GRAPHING SEVERAL POLAR EQUATIONS

General Form:

```
>plots[polarplot]({expr.1,expr.2,...,expr.n}, variable=a..b);
```

Example 32: On the same plane, sketch the graphs of $r = 3 + 2 \sin \theta$ and $r = \theta$.

```
>plots[polarplot]({3+2*sin(t), t}, t=0..2*Pi);
```



PART IX: VECTORS AND VECTOR OPERATIONS

Note: With Part IX, we begin the discussion of Maple applications that are uniquely relevant to Calculus III.

SKETCHING VECTORS

Sketching Vectors is part of the plots subroutine. If you have not already opened this subroutine during your current Maple session, input:


```
>with(plots):
```

Whether your vector is in two dimensions or three the format is the same:

General Form: `plots[arrow](vector, shape=arrow);`

Example 33: Sketch the vector $\langle -1, 3 \rangle$.

```
>plots[arrow](⟨-1, 3⟩, shape=arrow);
```




Alternately, you can assign the vector a name (notice the format below). Once it has that name you can refer to it from then on until you redefine the variable:

```
>v:=⟨-1, 3⟩;
>plots[arrow](v, shape=arrow);
```


Example 34: On the same plane, sketch the vectors $\mathbf{v} = \langle -1, 3 \rangle$, $\mathbf{u} = \langle 2, -1 \rangle$ and $\mathbf{v} + \mathbf{u}$. If you have several vectors to sketch, enter their names within square brackets []. Notice that you can enter a vector operation.

```
>u:=⟨2, -1⟩;
>plots[arrow]([v,u,v+u], shape=arrow);
```



Example 35: Sketch the vector $-\mathbf{i} - 4\mathbf{j} + 2\mathbf{k}$.

```
>plots[arrow] (<-1, -4, 2>, shape=arrow) ;
```



Click the “plot” drop down menu and select “axes” so you can see where your vector lies. I have chosen “normal” axes. Feel free to play with the other options in this menu.

VECTOR OPERATIONS

Dot Products

General Form: `>LinearAlgebra[DotProduct] (1st vector, 2nd vector) ;`

Example 36: Let $\mathbf{u} = \langle 2, 3, 4 \rangle$ and $\mathbf{v} = \langle -5, -4, -3 \rangle$. Find $\mathbf{u} \cdot \mathbf{v}$.

```
>u:=<2, 3, 4>:
```

```
>v:=<-5, -4, -3>:
```

```
>LinearAlgebra[DotProduct] (u, v) ;
```

-34

Cross Products

General Form: `>LinearAlgebra[CrossProduct] (1st vector, 2nd vector) ;`

Example 37: Using the same vectors find $\mathbf{u} \times \mathbf{v}$.

>LinearAlgebra[CrossProduct](u,v);

$$\begin{bmatrix} 7 \\ -14 \\ 7 \end{bmatrix}$$

Magnitude of a Vector

General Form: >VectorCalculus[Norm] (vector) ;

Example 38: Find the magnitude of the same vector $u = \langle 2,3,4 \rangle$.

>VectorCalculus[Norm] (u) ;
 $\sqrt{29}$

PART X: GRAPHING VECTOR FUNCTIONS


GRAPHING A SINGLE VECTOR FUNCTION IN THE PLANE

General Form: plot ([x-expression,y-expression,parameter=a..b]) ;

Example 39: Sketch the graph of the plane curve defined by

$$\vec{r}(t) = (3t^2 - 2t - 1)\vec{i} + t\vec{j}$$


>plot ([3*t^2-2*t-1, t, t=-2..3]) ;



GRAPHING SEVERAL VECTOR FUNCTIONS IN THE PLANE

General Form: plot ({ [1+t^2, t-2], [1+t^2, t-4] }, t=-2..4) ;

Example 40: On t-2)j.
 >plot ({ [1+t^2, t-2], [1+t^2, t-4] }, t=-2..4) ;



CURVES IN SPACE DEFINED BY VECTOR FUNCTIONS

General Form: `plots[spacecurve]([x-expr.,y-expr.,z-expr],var=a..b);`


Notice that the `]` is in a different location from where it was for graphing in the plane.

Example 41: Sketch the graph of the line defined by

$$\mathbf{r}(t) = (2 + 3t)\mathbf{i} + (5 + 4t)\mathbf{j} + (-3 + 2t)\mathbf{k}.$$


```
>plots[spacecurve]([2+3*t,5+4*t,-3+2*t],t=-4..5);
```

```
color="no coloring", axes="normal"
```



Example 42: Sketch the graph of the function $\mathbf{r}(t) = \sin t \mathbf{i} + \cos t \mathbf{j} + t \mathbf{k}$.

```
>plots[spacecurve]([sin(t),cos(t),t],t=-2*Pi..t=6*Pi);
```



PART XI: VECTOR CALCULUS

Note: These operations will require calling the Vector Calculus Subroutine. Type this first:

```
> with(VectorCalculus):
```

LIMITS

General Form: `>limit (vector function,parameter = value);`

Example 43: Find $\lim_{t \rightarrow 2} \left[\left(\frac{t^2 - 4}{t + 2} \right) \mathbf{i} + e^{(t-2)} \mathbf{j} \right]$

```
>limit (<(t^2-4)/(t+2), exp(t-2)>, t=2);
4e_x + e_y
```

What are those funky e things?

Maple uses an alternative notation for the standard unit vectors:

$$e_x = \mathbf{i}, \quad e_y = \mathbf{j}, \quad e_z = \mathbf{k}$$

DERIVATIVES

General Form: `>diff (vector function,parameter) ;`

Example 44: Find $\mathbf{r}'(t)$ if $\mathbf{r}(t) = \sqrt{t+3}\mathbf{i} + e^{4t}\mathbf{j} + \sin^2 3t\mathbf{k}$.

```
>diff (<sqrt(t+3), exp(4*t), (sin(3*t))^2>, t);
1/(2*sqrt(t+3))e_x + 4e^{4t}e_y + 6sin(3t)cos(3t)e_z
```

Example 45: Referring back to Example 44, find $\mathbf{r}'(6)$.

Use the “eval” command. You can use “%” to take the result of the previous calculation.

```
>eval (% , t=6);
1/18*sqrt(9)e_x + 4e^{24}e_y + 6sin(18)cos(18)e_z
```

INTEGRALS

Indefinite Integrals

General Form: $\text{>int}(\text{vector function, parameter}) ;$

Example 46: Evaluate $\int [t \sin(t^2)\mathbf{i} + \sec^2(4t)\mathbf{j}] dt.$

$\text{>int}(\langle t \sin(t^2), (\sec(4t))^2 \rangle, t) ;$

$$-\frac{1}{2} \cos(t^2)e_x + \frac{1}{4} \frac{\sin(4t)}{\cos(4t)}e_y$$

Definite Integrals

General Form: $\text{>int}(\text{vector function, parameter=a..b}) ;$

Example 47: Evaluate $\int_0^2 [t^2\mathbf{i} + t^3\mathbf{j} + t^4\mathbf{k}] dt$

$\text{>int}(\langle t^2, t^3, t^4 \rangle, t=0..2) ;$

$$\left(\frac{8}{3}\right)e_x + 4e_y + \left(\frac{32}{5}\right)e_z$$

ARC LENGTH

General Form: $\text{>ArcLength}(\text{vector function, parameter=a..b}) ;$

Example 48: Find the length of the curve C defined by the vector function

$$\mathbf{r}(t) = t\mathbf{i} + t^2\mathbf{j} \text{ on the interval } [0,5].$$

$\text{>ArcLength}(\langle t, t^2 \rangle, t=0..5) ;$

$$\frac{5}{2} \sqrt{101} - \frac{1}{4} \ln(-10 + \sqrt{101})$$


PART XII: GRAPHING IN THREE DIMENSIONS

GRAPHING FUNCTIONS OF TWO VARIABLES


General Form: `>plot3d (function,var.1=a..b,var.2=c..d) ;`

Example 49: Sketch the graph of $z = \sin x \cos y$.


`>plot3d(sin(x)*sin(y),x=-2*Pi..2*Pi,y=-2*Pi..2*Pi);`



Click on the graph and explore the options under “plot” pull down menu options “style”, “colors”, “axes”, “transparency” and “projections”. Here is the same diagram with axes = “boxed” and style = “patch and contour”. For this handout, color = “grayscale(Z)”.



Viewing Angles- One can “walk around” a three dimensional graph by changing the viewing angles. After clicking on the graph, change the values of θ and ϕ in the upper left hand corner of your screen **or** simply click on the graph and move the mouse around to change the viewing angles. θ = degrees counterclockwise from the positive x axis. ϕ = degrees downward from the positive z axis. Default: $\theta = \phi = 45^\circ$.





Example 50: Sketch the graph of $z = y^2 - x^2$.

`plot3d(y^2-x^2, x=-3..3, y=-3..3);`



with $\theta = 45^\circ$, $\phi = 45^\circ$:

with $\theta = 61^\circ$, $\phi = 61^\circ$:




with $\theta = 20^\circ$, $\phi = 45^\circ$:

with $\theta = 120^\circ$, $\phi = 105^\circ$:



If you'd like a contour map, click on your graph. In the “plot” pull down menu set the “style” to “contour” and change $\phi = 0^\circ$ so that you're looking down on the graph from directly overhead. Here is the contour map for Example 50.




GRAPHING SEVERAL FUNCTIONS OF TWO VARIABLES

General Form: `plot3d({expr.1,expr.2,...expr.n}, var.1=a..b,var.2=c..d);`

Example 51: Sketch the graphs of $z = 3x - 2y$ and $z = 8 \sin xy$.

`plot3d({3*x-2*y, 8*sin(x*y)}, x=-6..6, y=-6..6);`




THREE DIMENSIONAL SURFACES DEFINED PARAMETRICALLY

General Form: `plot3d([x-expr.,y-expr.,z-expr.], var.1=a..b,var.2=c..d);`

Example 52: Sketch the graph of the graph defined by

$$x = \cos s \cos t, \quad y = \cos s \sin t, \quad z = \sin s.$$

```
plot3d([cos(s)*cos(t), cos(s)*sin(t), sin(s)], s=0..2*Pi, t=0..2*Pi);
```



EQUATIONS IN THREE DIMENSIONS THAT DO NOT DEFINE FUNCTIONS

Make sure you have opened the “plots” subroutine since you started your Maple session.

```
with(plots):
```

General Form: `>implicitplot3d(equation,x=a..b,y=c..d,z=e..f);`


Example 53: Sketch the graph of $4x^2 + 9y^2 - z^2 = 36$.

```
implicitplot3d(4*x^2+9*y^2-z^2=36, x=-9..9, y=-9..9, z=-9..9);
```



Example 54: Sketch the graph of $9x^2 + 4z^2 = 36$.

```
>implicitplot3d(9*x^2+4*z^2=36,x=-2..2,y=-3..3,z=-5..5);
```



I set the transparency low and added axes.